

ESTATUAK ETA HIRIBURUAK

Geografia izan da oraingo programa honetan lantzeko aukeratu dugun gaia; Europako estatuak eta hiriburuak hain zuzen ere. Dena den, gaiari berari baino garrantzi handiagoa eman nahi diogu lantzeko moduari; gehienetan gertatzen den bezala programa bera beste arlo askotako gaiak lantzeko ere erabilgarri izan baitaiteke.

Zutabe batean estatuak eta bestean hiriburuak agertu ondoren, bi zutabe hauen arteko erlazioa aurkitzea izango da erabiltzaileak egin beharko duena, edo beste modu batera esanda, bikote unibokoak osatzea.

Pantailan agertuko dira une oro erabil daitezkeen teklak eta berauen funtzioak. Beraz, lanean hasteko prest baldin bazaude, aurrera!

```
10  REM ESTATUAK ETA HIRIBURUAK
20  KEY OFF:CLS:SCREEN 2:SCREEN 0
30  DIM TAU$(9,4),EST(31),KAP(9),NAH(9),ALE(9)
40  CLS:RANDOMIZE TIMER:KEY 10,"":KEY (10) ON KEY (10) GOSUB 900
50  YK=12:XK=20:COLOR 2:LER$="Kaixo, ongi etorria programa honetara!!!":
    GOSUB 1090
60  FOR I=1 TO 3000:NEXT I:CLS
70  FOR I=1 TO 31:EST(I)=I:NEXT I:KOP=31
80  FOR I=1 TO 9
90      A=INT(RND*KOP)+1:ALE(I)=EST(A)
100     FOR J=A TO KOP-1:EST(J)=EST(J+1):NEXT J
110     KOP=KOP-1
120  NEXT I
130  FOR I=1 TO 9:KAP(I)=I:NEXT I:KOP=9
140  RESTORE 1300
150  FOR I=1 TO 9
160     READ TAU$(I,4):B=INT(RND*KOP)+1:NAH(I)=KAP(B)
170     FOR J=B TO KOP-1:KAP(J)=KAP(J+1):NEXT J
180     KOP=KOP-1
190  NEXT I
200  FOR I=1 TO 9
210     RESTORE 1190
220     FOR J=1 TO ALE(I)
230         READ TAU$(I,1),TAU$(I,2):TAU$(I,3)="0"
240     NEXT J
250  NEXT I
260  RESTORE 1260
270  GOSUB 940:COLOR 3
280  FOR I=1 TO 9
290     LOCATE 4+2*I,10+INT((31-LEN(TAU$(I,1)))/2):PRINT TAU$(I,1)
300  NEXT I
310  FOR I=1 TO 9
320     LOCATE 4+2*I,55+INT((16-LEN(TAU$(NAH(I,2)))/2):PRINT TAU$(NAH(I,2)
330  NEXT I
340  ASMAT$="F":GAIZKI=0
350  WHILE ASMAT$="F" AND GAIZKI<5
360     P=1:WHILE TAU$(P,3)="1":P=P+1:WEND:TES$="" :ZUT=0
370     WHILE ASC(TES$)<>13
380         LOCATE 25,4:COLOR 11:PRINT CHR$(24);:COLOR 3:PRINT " :gora":
```

```

LOCATE 25,18:COLOR 11:PRINT CHR$(25);:COLOR 3:PRINT " behera":
LOCATE 25,33:COLOR 11:PRINT "<RETURN>";:COLOR 3:PRINT " hautatzeko"
390 LOCATE 25,59:COLOR 11:PRINT "<F10>";:COLOR 3:PRINT " amaitzeko"
400 IF ZUT=0 THEN LER$=TAU$(P,1) ELSE LER$=TAU$(NAH(P),2)
410 LOCATE 4+P*2,10+ZUT*45+INT((15*(2-ZUT)+1-LEN(LER$))/2):COLOR 27:
PRINT LER$
420 TE$=INKEY$:WHILE TE$<>"":TE$=INKEY$:WEND
430 TE$="":WHILE TE$="":TE$=INKEY$:WEND
440 IF ASC(TE$)<>13 AND ASC(RIGHT$(TE$,1))<>72 AND ASC(RIGHT$(TE$,1))<>80
THEN 430
450 IF (ZUT=0 AND TAU$(P,3)="0") OR (ZUT=1 AND TAU$(NAH(P),3)="0") THEN
COLOR 3 ELSE IF ZUT=0 THEN COLOR VAL(TAU$(P,4))
ELSE COLOR VAL(TAU$(NAH(P),4))
460 IF ZUT=0 THEN LER$=TAU$(P,1) ELSE LER$=TAU$(NAH(P),2)
470 LOCATE 4+P*2,10+ZUT*45+INT((15*(2-ZUT)+1-LEN(LER$))/2):PRINT LER$
480 SOUND 1000,.8
490 REM AUKERA
500 IF RIGHT$(TE$,1)<>"H" THEN 520
510 IF P>1 THEN P=P-1 ELSE BEEP:GOTO 540
520 IF RIGHT$(TE$,1)<>"P" THEN 540
530 IF P<9 THEN P=P+1 ELSE BEEP:GOTO 540
540 REM AMAUKERA
550 WEND
560 ZUT=ZUT+1
570 LOCATE 4+P*2,10+45*(ZUT-1)+INT((15*(3-ZUT)+1-LEN(LER$))/2):COLOR 11:
PRINT LER$
580 IF ZUT=2 THEN 600
590 EZP=P:P=1:TE$="" :GOTO 370
600 LOCATE 25,1:PRINT SPC(79):COLOR 3
610 IF EZP<>NAH(P) THEN 650
620 TAU$(EZP,3)="1"
630 FOR J4=1 TO 3:SOUND 1000*J4,.3*J4:NEXT J4:XK=5:YK=25:LER$=ONGI!!! "+
TAU$(EZP,1)+" estatuko hiriburua "+TAU$(NAH(P),2)+" da.":GOSUB 1090
640 K1=VAL(TAU$(EZP,4)):K2=K1:GOTO 680
650 SOUND 300,3:XK=5:YK=25:LER$='GAIZKI!!! "+TAU$(NAH(P),2)+" ez da "+
TAU$(EZP,1)+" estatuko hiriburua ':GAIZKI=GAIZKI+1:COLOR 3:GOSUB 1090
660 IF TAU$(EZP,3)="0" THEN K1=3 ELSE K1=VAL(TAU$(EZP,4))
670 IF TAU$(NAH(P),3)="0" THEN K2=3 ELSE K2=VAL(TAU$(NAH(P),4))
680 COLOR K1:LER$=TAU$(EZP,1):LOCATE 6+(EZP-1)*2,10+INT((31-LEN(LER$))/2):
PRINT LER$
690 COLOR K2:LER$=TAU$(NAH(P),2):LOCATE 6+(P-1)*2,55+INT((16-LEN(LER$))/2):
PRINT LER$
700 FOR I=1 TO 3000:NEXT I
710 ASMAT$="E"
720 FOR I=1 TO 9
730 IF TAU$(I,3)="0" THEN ASMAT$="F"
740 NEXT I
750 LOCATE 25,1:PRINT SPC(79)
760 WEND
770 COLOR 3
780 IF ASMAT$="E" THEN LOCATE 25,3:PRINT 'ONGI!!! Guztiak asmatu dituzu.:';
" Aurrera jarraitu nahi al duzu (B/E)?':GOTO 880
790 LOCATE 25,3:PRINT "Ez dituzu guztiak asmatu.:";
" Falta zitzaizkizunak ikusi nahi al dituzu (B/E)?'
800 TE$="" :WHILE TE$<>"B" AND TE$<>"b" AND TE$<>"E" AND TE$<>"e":TE$=INKEY$
:WEND
810 IF TE$="E" OR TE$="e" THEN 870
820 FOR I=1 TO 9:IF TAU$(NAH(I),3)="1" THEN 860
830 COLOR VAL(TAU$(NAH(I),4))
840 LOCATE 4+2*NAH(I),10+INT((31-LEN(TAU$(NAH(I),1)))/2):
PRINT TAU$(NAH(I),1)
850 LOCATE 4+2*I,55+INT((16-LEN(TAU$(NAH(I),2)))/2):
PRINT TAU$(NAH(I),2)
860 NEXT I
870 COLOR 3:LOCATE 25,1:PRINT SPC(79):LOCATE 25,26:
PRINT "Jarraitu nahi al duzu (B/E)?"

```

```

880  TE$=" "WHILE TE$<"B" AND TE$<"b" AND TE$<"E" AND TE$<"e":TE$=INKEY$:
      WEND
890  IF TE$="B" OR TE$="b" THEN 40
900  CLS:LOCATE 13,27:COLOR 2:PRINT "A G U R !!! Beste bat arte."
910  END
920  REM AMAIA ESTATUAK ETA HIRIBURUAK
930                                     REM
940  REM AZALPENA
950  READ KOD$
960  WHILE KOD$<"*"
970      IF KOD$="t" THEN READ XK,YK,KO,LER$:COLOR KO:GOSUB 1090:GOTO 1040
980      IF KOD$="te" THEN READ BUK,YK,KO,LER$:COLOR KO:GOSUB 1140:GOTO 1040
990      IF KOD$="r" THEN READ EZZ,GL,ESZ,BL,KO:COLOR KO:LOCATE GL,EZZ:
          PRINT CHR$(201);STRING$(ESZ-EZZ-1,205);CHR$(187):FOR J4=GL+1 TO BL-1:
          LOCATE J4,EZZ:PRINT CHR$(186):LOCATE J4,ESZ:PRINT CHR$(186):NEXT J4
1000     IF KOD$="r" THEN LOCATE BL,EZZ:PRINT CHR$(200);STRING$(ESZ-EZZ-1,205);
          CHR$(188)
1010     IF KOD$="re" THEN READ EZZ,GL,ESZ,BL,KO:COLOR KO:LOCATE GL,EZZ:
          PRINT CHR$(204);STRING$(ESZ-EZZ-1,205);CHR$(185):FOR J4=GL+1 TO BL-1:
          LOCATE J4,EZZ:PRINT CHR$(186):LOCATE J4,ESZ:PRINT CHR$(186):NEXT J4
1020     IF KOD$="re" THEN LOCATE BL,EZZ:PRINT CHR$(204);STRING$(ESZ-EZZ-1,205);
          CHR$(185)
1030     IF KOD$="rek" THEN READ EZZ,GL,ESZ,KO:COLOR KO:LOCATE GL,EZZ:
          PRINT CHR$(200):LOCATE GL,ESZ:PRINT CHR$(188)
1040     READ KOD$
1050  WEND
1060  RETURN
1070  REM AMAIA AZALPENA
1080                                     REM
1090  REM  IDATZI ESKUINERANTZ EKINTZA
1100     LOCATE YK,XK:FOR J4=1 TO LEN(LER$):PRINT MID$(LER$,J4,1);
          FOR J=1 TO 40:NEXT J:SOUND 50,1:NEXT J4
1110     RETURN
1120  REM AMAIA IDATZI ESKUINERANTZ EKINTZA
1130                                     REM
1140  REM  IDATZI EZKERRERANTZ EKINTZA
1150     FOR J4=1 TO LEN(LER$):LOCATE YK,BUK-J4:PRINT LEFT$(LER$,J4):
          FOR J=1 TO BUK-J4:NEXT J:NEXT J4
1160     RETURN
1170  REM  AMAIA IDATZI EZKERRERANTZ EKINTZA
1180                                     REM
1190  REM  ESTATUAK ETA HIRIBURUAK
1200     DATA "Albania","Tirana","Andorra","Andorra La Vella","Austria",
"Vienna","Belgika","Brussela","Bulgaria","Sofia","Txekoslovakia","Praga",
"Danimarka","Kopenhage","Finlandia","Helsinki","Frantzia","Paris","Aleman Errepublikak Demokratikoa"
1210     DATA "Berlin","Aleman Errepublikak Federala","Bonn","Grezia","Athenas",
"Hungaria","Budapest","Islandia","Reykjavik","Irlanda, Eire","Dublin","Italia",
"Erroma","Liechtenstein","Vaduz","Luxemburg","Luxemburg","Malta","Valletta"
1220     DATA "Herbehereak","Amsterdam","Norvegia","Oslo","Polonia",
"Varsovia","Portugal","Lisboa","Errumania","Bukarest","San Marino","San
Marino","S.E.S.B.,"Mosku","Espainia","Madril","Suedia","Stockholm","Suitza",
"Berna"
1230     DATA "Erresuma batua","Londres","Jugoslavia","Belgrad"
1240  REM AMAIA ESTATUAK ETA HIRIBURUAK
1250                                     REM
1260  REM  DATUAK
1270     DATA "t",52,1,2,"ESTATUAK ETA HIRIBURUAK","t",29,2,3,"=====
1280     DATA "t",22,4,11,"ESTATUAK","r",8,5,42,7,5,"re",8,7,42,9,5,"re",8,9,42,
11,5,"re",8,11,42,13,5,"re",8,13,42,15,5,"re",8,15,42,17,5,"re",8,17,42,19,5,"re",
8,19,42,21,5,"re",8,21,42,23,5,"rek",8,23,42,5
1290     DATA "t",59,4,11,"HIRIBURUAK","r",53,5,72,7,5,"re",53,7,72,9,5,"re",53,
9,72,11,5,"re",53,11,72,13,5,"re",53,13,72,15,5,"re",53,15,72,17,5,"re",53,17,72,
19,5,"re",53,19,72,21,5,"re",53,21,72,23,5,"rek",53,23,72,5,"*"
1300     DATA "2","4","7","6","15","10","12","13","14"
1310  REM AMAIA DATUAK

```

Programaren azalpena

Programaren azalpena hasteko, kasu honetan intergarri iruditu zaigu erabiltzen diren taulak zeintzuk diren eta zertarako erabiltzen diren adieraztea.

TAU\$ (9,4): Bertan, pantailan agertuko diren 9 estatuen datuak gordeko dira. TAU\$ (I,1)en estatua bera gordeko da; TAU\$ (I,2)n estatuko hiriburua, TAU\$ (I,3)k bikote hori jadanik asmatu den (1) ala ez (Ø) adieraziko du eta TAU\$ (I,4) osagaiak bikoteari dagokion kolorea.

EST (31): 1etik 31ra bitarteko zenbakiak gordeko dira taula honetan; 31 estatu bait daude aukeratu ahal izateko.

ALE (9): EST taulatik aukeratutako 9 zenbaki gordeko dira taula honetan.

KAP (9): 1etik 9ra bitarteko zenbakiak gordeko dira.

NAH (9): Aurreko taulako zenbakiak nahasteko erabiliko da. Hots, NAH (I) elementuak ezkerreko zutabean I. posizioan agertuko den estatuko hiriburua eskuineko zutabean zein posizio okupatuko duen adieraziko du.

Goazen bada hau ikusi ondoren programa bera aztertzeraz:

10-60: Programaren hasieraketa.

70-120: 31 estatuetatik 9 hartuko dira aleatorioki.

130-190: Estatu/hiriburu bikote bakoitzeko kolore bat eta hiriburua eskuineko zutabean zein posizioan agertuko den definituko da.

200-250: Aukeratutako datuak irakurri besterik ez da egiten.

260-330: Hasierako pantaila irudikatzen da. Estatuak ezkerreko zutabean eta hiriburua eskuinekoan nahasturik, guztiak kolore berdinez.

350-760: Guztiak asmatzen ez dituen bitartean eta 5 akats egiten ez dituen bitartean, erabiltzaileak lehenbizi ezkerreko zutabetik bat aukeratu beharko du. Horretarako ↑ eta ↓ teklak erabili ahal izango ditu, aukeratu duen elementuan kokatu ondoren <RETURN> sakatu beharko duelarik. (370-550) Estatua hautatu duenean automatikoki pasatuko da eskuineko zutabera, hiriburuen zutabera eta hemen gauza bera egin beharko du. (560-600)

Aukeraketa zuzena izan bada, bikoteari dagokion koloreaz bereiztuko dira, bai estatua era bai hiriburua. (620-640). Asmatu ez bada, berriro ere lehengo kolorean geratuko dira eta 5. akatsa izan ez bada behintzat beste saio bat egiteko aukera eskainiko da (650-750).

780-910: Jokoa bikote guztiak aurkitu direlako edota guztiak aurkitu aurretik 5 akats egin direlako amaituko da. Bigarren kasuan, emaitza ikus-teko aukera ere ematen da. (820-860).

940-1070: Hasierako pantaila, hau da, errekoadroak etab. irudikatzeko erabiltzen den azpirrutina da hau.

1090-1120: Karaktereak banaka eta soinua aterata idazteko azpirrutina.

1140-1170: Karaktere-kateak eskuinaldetik ezker aldera idazteko azpirrutina.

1190-1240: Programa nagusiko datuak, hots, estatuak eta hiriburua.

1260-1310: Hasierako pantaila egiteko erabiltzen diren kode eta datuak. ■

**EUSKAL
ADITZ
LAGUNTZAILEA
(Softwarea)**

*Ordenadorea erabiliz,
aditzarekin jolastuz, milaka
ariketa egin ditzakezu*

**JADANIK
KALEAN**